

Lawrence Berkeley National Laboratory

Recent Work

Title

UCB-NE-107 USER'S MANUAL

Permalink

<https://escholarship.org/uc/item/94z4z51m>

Author

Lee, M.W.L.

Publication Date

1989-03-01



Lawrence Berkeley Laboratory
UNIVERSITY OF CALIFORNIA

EARTH SCIENCES DIVISION

RECEIVED
REFERENCE
BERKELEY LABORATORY

SEP 14 1989

LIBRARY AND
DOCUMENTS SECTION

UCB-NE-107 User's Manual

W.W.-L. Lee

March 1989

For Reference

Not to be taken from this room



DISCLAIMER

This document was prepared as an account of work sponsored by the United States Government. While this document is believed to contain correct information, neither the United States Government nor any agency thereof, nor the Regents of the University of California, nor any of their employees, makes any warranty, express or implied, or assumes any legal responsibility for the accuracy, completeness, or usefulness of any information, apparatus, product, or process disclosed, or represents that its use would not infringe privately owned rights. Reference herein to any specific commercial product, process, or service by its trade name, trademark, manufacturer, or otherwise, does not necessarily constitute or imply its endorsement, recommendation, or favoring by the United States Government or any agency thereof, or the Regents of the University of California. The views and opinions of authors expressed herein do not necessarily state or reflect those of the United States Government or any agency thereof or the Regents of the University of California.

**LBL-26672
UCB-NE-4132
UC-70**

UCB-NE-107

User's Manual

William W.-L. Lee

**Earth Sciences Division
Lawrence Berkeley Laboratory
1 Cyclotron Road
Berkeley, CA 94720**

March 1989

Work supported by the Director, Office of Civilian Radioactive Waste Management, Office of System Integration and Regulation, Licensing and Compliance Division, of the U. S. Department of Energy under contract DE-AC03-76SF00098.

INTRODUCTION

The purpose of this manual is to provide users of UCB-NE-107 with the information necessary to use UCB-NE-107 effectively.

DESCRIPTION OF THE CODE

UCB-NE-107¹ is a computer code for calculating the fractional release rate² of readily soluble radionuclides that are released from nuclear waste emplaced in water-saturated porous media. Waste placed in such environments will gradually dissolve. For many species such as actinides and rare earths, the process of dissolution is governed by the exterior flow field, and the chemical reaction rate or leaching rate.³ However, for readily soluble species such as ¹³⁵Cs, ¹³⁷Cs and ¹²⁹I, it has been observed that their dissolution rates are rapid. UCB-NE-107 is a code for calculating the release rate at the waste/rock interface, to check compliance with the U. S. Nuclear Regulatory Commission's (USNRC) subsystem performance objective. It is an implementation of the analytic solution given below.

THEORY

Small fractions of the more soluble species, e.g., cesium and iodine, are released from the UO₂ matrix during reactor operation and are present in UO₂ grain boundaries and pores and in fuel-cladding gaps. Upon contact with ground water that penetrates failed containers and fills voids in waste packages, these readily soluble species rapidly dissolve. Also there is continued slow dissolution of these same soluble species as the waste matrix dissolves in the void water. We assume that the surrounding porous rock is in direct contact with the well-mixed void liquid (Figure 1), that contains initially a specified quantity of these readily soluble species. The species migrate into the porous material under a concentration gradient. Assuming that advective transport in the near-field is small compared with diffusive transport, the space-time-dependent concentration $N(x, t)$ in pore water is given by

$$K \frac{\partial N(x, t)}{\partial t} = D_f \frac{\partial^2 N(x, t)}{\partial x^2} - \lambda K N(x, t), \quad x > a, \quad t > 0 \quad (1)$$

where D_f is the species diffusion coefficient [L^2/T] in the porous rock, and K is the species retardation coefficient.

The initial and boundary conditions are

$$N(x, 0) = 0, \quad x > a \quad (2)$$

$$N(\alpha, t) = N(t), \quad t \geq 0 \quad (3)$$

$$N(\infty, t) = 0, \quad t \geq 0 \quad (4)$$

where α is the thickness of the "gap" filled with void water. Here $N(t)$ is the time-dependent well-mixed concentration of the soluble species in the water in the void. To solve for $N(t)$, the mass balance in the void is

$$V \frac{dN(t)}{dt} = \dot{m}_f(t) - \dot{m}(t) - \lambda V N(t), \quad t > 0 \quad (5)$$

where $\dot{m}_f(t)$ is the rate of dissolution of the species from the waste matrix into the void volume V , and $\dot{m}(t)$ is the rate of species migration into the rock. To solve (5), we use the initial condition

$$N(0) = N^0 \quad (6)$$

where N^0 is the initial concentration of the species in the void water.

Chambré⁴ has obtained the solution to (1) through (6) resulting in the concentration of soluble species in the void as

$$N(t) = N^0 e^{-\lambda t} F(\beta^2 t) + \frac{1}{V} \int_0^t \dot{m}_f(t - \tau) e^{-\lambda \tau} F(\beta^2 \tau) d\tau, \quad t > 0 \quad (7)$$

where

$$F(\beta^2 t) \equiv e^{\beta^2} \operatorname{erfc} \sqrt{\beta^2 t}$$

and

$$\beta \equiv \sqrt{D_f K \epsilon^2 / \alpha^2}$$

The mass flux of the species into the rock is

$$\dot{m}(t) = -SD_f \epsilon \frac{\partial N(\alpha, t)}{\partial x}, \quad t > 0 \quad (8)$$

where S is the surface area of the interface between the void space and the rock ($S \equiv V/\alpha$). Using (6) the solution to (7) is

$$\dot{m}(t) = N^0 \beta V e^{-\lambda t} \left\{ \frac{1}{\sqrt{\pi t}} - \beta F(\beta^2 t) \right\} + \beta \int_0^t \dot{m}_f(t - \tau) e^{-\lambda \tau} \left\{ \frac{1}{\sqrt{\pi \tau}} - \beta F(\beta^2 \tau) \right\} d\tau, \quad t > 0 \quad (9)$$

from which the fractional release rate can be calculated.

ALGORITHM

UCB-NE-107 is an implementation of (9). Eq. (9) shows that the flux rate is the sum of the flux from the species released instantaneously in the gap, and the species released congruently from the waste matrix. UCB-NE-107 computes the two fluxes as well as the total flux, plus the fractional release rate, which is the total flux divided by the initial or 1000-year inventory. UCB-NE-107 also computes the concentration in the gap due to the initially released slug, and that from congruent release, as well as their sum.

The evaluation of

$$F(\beta^2 t) \equiv e^{\beta^2} \operatorname{erfc} \sqrt{\beta^2 t}$$

is by an adaptation of the rational approximation of the error function.⁵

Numerical Illustration

Figure 2 shows the fractional release rates of ^{135}Cs , ^{137}Cs and ^{129}I based on (9). In this illustration we consider the release from a bare waste form exposed to ground water shortly after emplacement, conservatively neglecting the mass-transfer resistance due to containers and cladding. The lower curves show the solubility-limited dissolution rate from the waste matrix, assuming congruent dissolution and a uranium solubility of 10^{-3} g/m³. A conservatively high diffusion coefficient of 0.12 m²/a is used throughout the analysis, and a porosity of 0.01 is assumed. The fractional release rates in Figure 2 are normalized to initial inventories. At $t = 0$, one percent of these species is assumed to dissolve into the water-filled void space of 0.45 m³. The calculated U. S. Nuclear Regulatory Commission fractional release rate limits on these species, based on initial inventories, are

$$^{135}\text{Cs} = 4.5 \times 10^{-5} \text{ per year}$$

$$^{137}\text{Cs} = 1.8 \times 10^{-10} \text{ per year}$$

$$^{129}\text{I} = 5.4 \times 10^{-4} \text{ per year}$$

With these parameters, the early contribution from gap inventory is more than 10^7 -fold greater than the contribution from the waste matrix. For early times, the fractional release rates of ^{135}Cs and ^{137}Cs are almost equal, but the ^{137}Cs release rate decreases rapidly because of radioactive decay, whereas the release rate of 3×10^6 -year ^{135}Cs is characteristic of a stable species. Because of the assumed negligible sorption of iodine in the rock, the early fractional release rate of 1.7×10^7 -year ^{129}I is lower than the rates of strongly sorbing cesium, but the less rapid early depletion of iodine in the void water results in a greater fractional release rate of iodine after about 20 years.

The fractional release rate of ^{137}Cs exceeds the USNRC release rate limit for several hundred years. The presence of metallic containers, and the time-distributed nature of container failure will likely help meet the requirement.

INPUT

The input consists of 4 lines. UCB-NE-107 looks for a file called UCB107.dat as the input.

First line (free format)

RSP; equivalent waste-sphere radius (m)
DF; diffusion coefficient (m^2/a)
EPS; porosity of rock (-)
VOL; volume of gap (m^3)
WID; gap width (m)

Second line (free format)

AK; retardation coefficient of species (-)
CONI; initial inventory of species in waste solid (g)
SNGAP; initial concentration of species in gap (g/m^3)
RAMBDA; decay constant of species ($1/\text{a}$)

Third line (free format)

ZK; retardation coefficient of the waste matrix (-)
CONA; initial inventory of matrix in the waste (g)
CS; solubility of dissolution controller (g/m^3)

Fourth line (free format)

NOPTION; option for output to FOR009
TJUG; which initial inventory to use

An example input file follows.

```
.752, 3.1536d-4, .001, .45, .1,  
10., 1.38d3, 30.76, 2.31d-7,  
20., 5.d6, 1.d-3,  
2, 0.,
```

OUTPUT

There are two output streams. The first is a tabular output, currently directed to the screen. The second is a stream of output data, for use by a X-Y plotter or plotting program such as TELL-A-GRAF.

The tabular output is currently directed to the screen by a

WRITE (*,xxx)

statement. In a DEC/VAX computer this output can be directed to a file by a command such as

ASSIGN SYSS\$OUTPUT filename

Please note that the tabular output is 80 columns wide. The output first prints out most of the input data. It also prints out the calculated leach time for the waste. Then in a table of 8 columns it lists

Time (a)

Concentration in the Gap Due to Instantaneous Release (g/m^3)

Concentration in the Gap Due to Congruent Release from the Matrix (g/m^3)

Total Concentration in the Gap (g/m^3)

Mass Flux Rate at Waste/Rock Interface Due to Instantaneous Release (g/a)

Mass Flux Rate at Waste/Rock Interface Due to Congruent Release (g/a)

Total Mass Flux Rate at Waste/Rock Interface (g/a)

Fractional Release Rate at Waste/Rock Interface ($1/\text{a}$)

The file FOR009.dat contains two columns of numbers. The first column is always the time in years. The second column depends on the value of NOPTION. If NOPTION=1, the second column contains the total fractional release rate (1/a). If NOPTION=2, the second column contains the fractional release rate (FRR) considering only the contribution of the instantaneously released species. If NOPTION=3, the second column contains the fractional release rate (FRR) considering the contribution from congruent dissolution from the waste matrix only. If NOPTION=4, the second column contains the total mass flux (g/a). If NOPTION=1, 2, 3 the fractional release rate has been divided by an inventory. If TJUG=0 : the fractional release rate is based on the initial inventory and if TJUG=1000, the fractional release rate is based on the thousand-year inventory.

RESOURCE REQUIREMENTS

UCB-NE-107 is written in FORTRAN and does not call any outside libraries. It operates on DEC/VAX and DEC/UNIX machines.

VERIFICATION & BENCHMARKING

UCB-NE-107 is an implementation of (9). It was verified through hand calculations. The correctness of UCB-NE-107 was also checked this way. Eq. (9) was given to several workers and they were asked to write computer programs independently. The outputs of the various programs are then compared or benchmarked. They gave identical results. The current UCB-NE-107 is in fact a combination of the best features of several programs.

REFERENCES

1. C. L. Kim, P. L. Chambré and T. H. Pigford, "Mass-Transfer Limited Release of a Soluble Waste Species," *Trans. Am. Nuc. Soc.*, 52, 80, 1986.
2. U. S. Nuclear Regulatory Commission, Disposal of High-Level Radioactive Wastes in Geologic Repositories, 10 *Code of Federal Regulations* 60.113(a)(1)(ii)(B).
3. P. L. Chambré, C. H. Kang, W. W.-L. Lee and T. H. Pigford, "The Role of Chemical Reaction in Waste-Form Performance," in M. J. Apted and R. E. Westerman (eds.), *Scientific Basis for Nuclear Waste Management XI*, 285, Pittsburgh, Materials Research Society, 1988
4. P. L. Chambré, to be published.
5. K. H. Haskell and R. E. Jones, 1978. *Brief Instructions for Using the Sandia Mathematical Subroutine Library (Version 7.2)*, SAND77-1441.





Figure 1. Release of Soluble Species

Figure 2. An Example of FOR009 Output Plotted Using TELL-A-GRAF.
Fractional Release Rates of Some Soluble Species.



APPENDIX A: CODE LISTING

```

c PROGRAM
c *****
c
c Gap Problem without Surface Reaction
c - in one dimensional geometry
c (x<0 - waste solid ; 0<x<a - gap ; x>a - rock)
c - gap is filled with water
c - concentration in the gap is uniform due to sufficient stirring
c - congruent dissolution from waste solid
c (solubility-limited diffusion-controlled dissolution)
c - diffusive mass transfer in the saturated porous rock
c
c >>Input<<
c First line (free format)
c RSP; equivalent waste-sphere radius (m)
c DF; diffusion coefficient (m**2/yr)
c EPS; porosity of rock (-)
c VOL; volume of gap (m**3)
c WID; gap width (m)
c Second line (free format)
c AK ; retardation coefficient of species (-)
c CONI; initial inventory of species in waste solid (g)
c SNGAP; initial concentration of species in gap (g/m**3)
c RAMBDA; decay constant of species (1/yr)
c Third line (free format)
c ZK ; retardation coefficient of dissolution controller (-)
c CONA ; initial inventory of controller in waste (g)
c CS ; solubility of dissolution controller (g/m**3)
c Fourth line (free format)
c if NOPTION=1 : write to FOR009 the total fractional release rate
c =2 : fractional release rate from gap only
c =3 : fractional release rate from solid only
c =4 : the total mass flux [g/yr]
c it TJUG=0 : based on initial inventory
c =1000 : thousand year inventory
c
c >>Output<<
c gap concentration of specie as a function of time
c mass flux from gap into surrounding porous rock
c
c Author --- Chang-Lak Kim
c Date --- Sep. 24, 1985
c Revision - Oct. 18, 1985
c Revised by Yongsoo Hwang July, 31 1987
c Second Revision - September, 23 1988, DERFL instead of DAWSON
c Third Revision - January 1989 (fsolid=0 after tleach)
c Fourth Revision - January 1989 by W. Lee, Output format
c
c Input File must be called UCB107.DAT
c 80-column output to screen unless SYS$OUTPUT assigned
c Output for TELL-A-GRAF are in FOR009 VIA THE OPTIONS
c FOR009 output in FRACTIONAL RELEASE RATES (FRR)
c *****
c IMPLICIT DOUBLE PRECISION (A-H,O-Z)
c EXTERNAL DAWSON,SDAWSON
c COMMON /UCB/ PI
c OPEN(5,FILE='UCB107.DAT',STATUS='UNKNOWN')
c READ(5,*) RSP,DF,EPS,VOL,WID
c READ(5,*) AK,CONI,SNGAP,RAMBDA

```

```

      READ(5,*) ZK,CONA,CS
      READ(5,*) NOPTION,TJUG

c
c
c      internal parameters
c
      PI=DACOS(-1D0)
      BETA=DSQRT(DF*AK*EPS*EPS/(WID*WID))
      COEF=CS*CONI*DSQRT(ZK/AK)/CONA

c
c
c      matrix leach time calculation
c
      revision 2
      TLEACH is calculated based on the equation (2.32) in KIM's thesis
c      A1 is the initial controller mass
      A1=CONA
      A2=4.0d0*PI*EPS*RSP*DF*CS
      AA=A1/A2
      B1=8.0d0*EPS*CS*RSP*RSP*DSQRT(PI*DF*ZK)
      B2=B1*B1-B1*DSQRT(B1*B1+4.0d0*A2*A1)
      BB=B2/(2.0d0*A2*A2)
      TLEACH=AA+BB

c
c      writing out input parameters
c
      write(*,600) rsp,df,wid,vol,eps
600 format(24h INPUT DATA
      *      24h Waste sphere radius      =, d9.3, 2h m/
      *      24h diffusion coefficient    =,1pd9.3,7h m**2/s/
      *      24h gap width                =,d9.3,2h m/
      *      24h volume of gap            =,d9.3,5h m**3/
      *      24h porosity                 =,d9.3/)
      write(*,601) ak,coni,sngap,rambda
601 format(52h retardation coefficient of the radionuclide      =,
      *      1pd9.3/
      *      52h initial inventory of the radionuclide in waste =,
      *      d9.3,2h g/
      *      52h initial concentration of the radionuclide in gap =,
      *      d9.3,7h g/m**3/
      *      52h decay constant of the radionuclide              =,
      *      d9.3,4h 1/s/)
      write(*,602) zk,cona,cs,tleach
602 format(52h retardation coefficient of the controller      =,
      *      1pd9.3/
      *      52h initial inventory of the controller in waste =,
      *      d9.3,2h g/
      *      52h solubility limit of the controller            =,
      *      d9.3,7h g/m**3/
      *      52h matrix leach time                              =,
      *      d9.3, 2h s/)
      WRITE (*,606)
606 FORMAT(' TIME CONCENTRATION FROM ',
      *      FLUX FROM')
      WRITE (*,603)
603 format(' years gap matrix TOTAL gap ',
      *      27h matrix TOTAL FRR/1x,80(1h-))

```

```

c      beginning of calculations
c
      IF(TLEACH.GE.1.0d7) GO TO 231
      TDEAD=TLEACH
      go to 232
231  TDEAD=1.0d7
232  CONTINUE
      ALL=1.d0
      SUP=-1.d0
110  TSE=ALL*10.0d0**SUP
c
      RT=DEXP(-RAMBDA*TSE)
      BT=BETA*BETA*TSE
      CGAP=SNAP*DAWSON(BT)*RT
      CALL SDAWSON(BETA,TSE,ANSWER)
      FGAP=BETA*VOL*SNAP*RT*ANSWER
      CSOLID=COEF*(1.-DAWSON(BT))*RT
      IF(TSE.GT.TLEACH) GO TO 45671
      FSOLID=BETA*BETA*VOL*COEF*RT*DAWSON(BT)
      GO TO 45672
45671 FSOLID=0.0d0
45672 CONTINUE
      CN=CGAP+CSOLID
      IF(NOPTION.EQ.4) GO TO 331
      IF(NOPTION.EQ.1) GO TO 331
      IF(NOPTION.EQ.2) GO TO 332
      FLUX=FSOLID
      GO TO 333
332  FLUX=FGAP
      GO TO 333
331  FLUX=FSOLID+FGAP
      IF(NOPTION.EQ.4) GO TO 934
      GO TO 333
934  WRITE(9,*) TSE,FLUX
      GO TO 590
333  CONTINUE
      SONE=RAMBDA*TJUG
      FINI=DEXP(-SONE)*CONI
      FRAC=FLUX/FINI
      WRITE(9,*) TSE,FRAC
      WRITE(*, 604) TSE,CGAP,CSOLID,CN,FGAP,FSOLID,FGAP+FSOLID,
      * (FGAP+FSOLID)/FINI
590  CONTINUE
      IF(TSE.GE.TDEAD) GO TO 120
      IF(ALL.GE.9.0d0) GO TO 130
      ALL=ALL+1.0d0
604  format(1pd10.3,7d10.2)
      GO TO 110
130  ALL=1.0d0
      SUP=SUP+1.0d0
      GO TO 110
120  STOP
      END
c
c
c
c
c
      REAL*8 FUNCTION DAWSON(X)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)

```

```

EXTERNAL ERFL
XXZ=DSQRT(X)
DAWSON=ERFL(XXZ)
RETURN
END

```

```

C
SUBROUTINE SDAWSON(BETA,TSE,ANSWER)
IMPLICIT DOUBLE PRECISION (A-H,O-Z)
EXTERNAL ERFL
COMMON /UCB/ PI
CCX=BETA*BETA*TSE
CCX=DSQRT(CCX)
ANSWER=1.0d0/DSQRT(PI*TSE)-BETA*ERFL(CCX)
RETURN
END

```

```

C
FUNCTION ERFL(XS)
IMPLICIT DOUBLE PRECISION (A-H,O-Z)
REAL*8 P(6),Q(6),R(4),S(4)
DATA P/22.898992851859D0,26.094746956075D0,14.571898596926D0,
*4.2677201070898D0,.56437160686381D0,-6.0858151959688D-6/,
*Q /22.898985749891D0,51.933570687552D0,50.273202863803D0,
*26.288795758761D0,7.5688482293618D0,1D0/,
*R /-1.21308276389978D-2,-.119903955268146D0,-.243911029488626D0,
*-3.24319519277746D-2/,
*S /4.3002664345277D-2,.489552441961437D0,1.43771227937118D0,1D0/
DATA SQPI/.564189583547756D0/
AX=DABS(XS)
IF (AX.GT.4.0) GO TO 10
TMP=P(1)+AX*(P(2)+AX*(P(3)+AX*(P(4)+AX*(P(5)+AX*(P(6))))))
A=TMP/(Q(1)+AX*(Q(2)+AX*(Q(3)+AX*(Q(4)+AX*(Q(5)+AX*(Q(6)))))
ERFL=DSIGN(A,XS)
GO TO 11
10 CONTINUE
X2=1./AX/AX
B=X2*(R(1)+X2*(R(2)+X2*(R(3)+X2*(R(4)))/(S(1)+X2*(S(2)+X2*
*(S(3)+X2*(S(4))))))
A=(SQPI+B)/AX
ERFL=DSIGN(A,XS)
11 CONTINUE
RETURN
END

```

APPENDIX B: SAMPLE OUTPUT

*Output from GAP

Jan. 16, 1989

INPUT DATA

Waste sphere radius =0.752D+00 m
 diffusion coefficient =3.154D-04 m**2/a
 gap width =1.000D-01 m
 volume of gap =4.500D-01 m**3
 porosity =1.000D-03

 retardation coefficient of the radionuclide =1.000D+01
 initial inventory of the radionuclide in waste =1.380D+03 g
 initial concentration of the radionuclide in gap =3.076D+01 g/m**3
 decay constant of the radionuclide =2.310D-07 1/a

 retardation coefficient of the controller =2.000D+01
 initial inventory of the controller in waste =5.000D+06 g
 solubility limit of the controller =1.000D-03 g/m**3
 matrix leach time =1.678D+15 a

TIME	CONCENTRATION FROM			FLUX FROM			
years	gap	matrix	TOTAL	gap	matrix	TOTAL	FRR
1.000D-01	3.08D+01	7.81D-11	3.08D+01	1.39D-02	5.54D-14	1.39D-02	1.00D-05
2.000D-01	3.08D+01	1.10D-10	3.08D+01	9.80D-03	5.54D-14	9.80D-03	7.10D-06
3.000D-01	3.07D+01	1.35D-10	3.07D+01	8.00D-03	5.54D-14	8.00D-03	5.80D-06
4.000D-01	3.07D+01	1.56D-10	3.07D+01	6.93D-03	5.54D-14	6.93D-03	5.02D-06
5.000D-01	3.07D+01	1.75D-10	3.07D+01	6.20D-03	5.54D-14	6.20D-03	4.49D-06
6.000D-01	3.07D+01	1.91D-10	3.07D+01	5.66D-03	5.54D-14	5.66D-03	4.10D-06
7.000D-01	3.07D+01	2.07D-10	3.07D+01	5.24D-03	5.54D-14	5.24D-03	3.80D-06
8.000D-01	3.07D+01	2.21D-10	3.07D+01	4.90D-03	5.54D-14	4.90D-03	3.55D-06
9.000D-01	3.07D+01	2.34D-10	3.07D+01	4.62D-03	5.54D-14	4.62D-03	3.35D-06
1.000D+00	3.07D+01	2.47D-10	3.07D+01	4.38D-03	5.54D-14	4.38D-03	3.17D-06
2.000D+00	3.07D+01	3.49D-10	3.07D+01	3.10D-03	5.53D-14	3.10D-03	2.24D-06
3.000D+00	3.07D+01	4.28D-10	3.07D+01	2.53D-03	5.53D-14	2.53D-03	1.83D-06
4.000D+00	3.07D+01	4.94D-10	3.07D+01	2.19D-03	5.53D-14	2.19D-03	1.59D-06
5.000D+00	3.07D+01	5.52D-10	3.07D+01	1.96D-03	5.53D-14	1.96D-03	1.42D-06
6.000D+00	3.07D+01	6.05D-10	3.07D+01	1.79D-03	5.53D-14	1.79D-03	1.29D-06
7.000D+00	3.07D+01	6.53D-10	3.07D+01	1.65D-03	5.53D-14	1.65D-03	1.20D-06
8.000D+00	3.07D+01	6.98D-10	3.07D+01	1.55D-03	5.53D-14	1.55D-03	1.12D-06
9.000D+00	3.07D+01	7.41D-10	3.07D+01	1.46D-03	5.53D-14	1.46D-03	1.06D-06
1.000D+01	3.07D+01	7.81D-10	3.07D+01	1.38D-03	5.53D-14	1.38D-03	1.00D-06
2.000D+01	3.07D+01	1.10D-09	3.07D+01	9.76D-04	5.52D-14	9.76D-04	7.07D-07
3.000D+01	3.07D+01	1.35D-09	3.07D+01	7.96D-04	5.52D-14	7.96D-04	5.77D-07
4.000D+01	3.06D+01	1.56D-09	3.06D+01	6.89D-04	5.52D-14	6.89D-04	4.99D-07
5.000D+01	3.06D+01	1.74D-09	3.06D+01	6.16D-04	5.51D-14	6.16D-04	4.46D-07
6.000D+01	3.06D+01	1.91D-09	3.06D+01	5.62D-04	5.51D-14	5.62D-04	4.07D-07
7.000D+01	3.06D+01	2.06D-09	3.06D+01	5.20D-04	5.51D-14	5.20D-04	3.77D-07
8.000D+01	3.06D+01	2.20D-09	3.06D+01	4.86D-04	5.51D-14	4.86D-04	3.52D-07
9.000D+01	3.06D+01	2.34D-09	3.06D+01	4.58D-04	5.51D-14	4.58D-04	3.32D-07
1.000D+02	3.06D+01	2.46D-09	3.06D+01	4.34D-04	5.50D-14	4.34D-04	3.15D-07
2.000D+02	3.05D+01	3.47D-09	3.05D+01	3.06D-04	5.49D-14	3.06D-04	2.22D-07
3.000D+02	3.04D+01	4.25D-09	3.04D+01	2.49D-04	5.48D-14	2.49D-04	1.80D-07
4.000D+02	3.04D+01	4.90D-09	3.04D+01	2.15D-04	5.47D-14	2.15D-04	1.56D-07
5.000D+02	3.03D+01	5.47D-09	3.03D+01	1.92D-04	5.46D-14	1.92D-04	1.39D-07
6.000D+02	3.03D+01	5.98D-09	3.03D+01	1.75D-04	5.45D-14	1.75D-04	1.27D-07
7.000D+02	3.02D+01	6.46D-09	3.02D+01	1.61D-04	5.45D-14	1.61D-04	1.17D-07
8.000D+02	3.02D+01	6.90D-09	3.02D+01	1.51D-04	5.44D-14	1.51D-04	1.09D-07
9.000D+02	3.02D+01	7.31D-09	3.02D+01	1.42D-04	5.43D-14	1.42D-04	1.03D-07
1.000D+03	3.01D+01	7.70D-09	3.01D+01	1.34D-04	5.43D-14	1.34D-04	9.74D-08
2.000D+03	2.99D+01	1.08D-08	2.99D+01	9.38D-05	5.38D-14	9.38D-05	6.80D-08

3.000D+03	2.97D+01	1.32D-08	2.97D+01	7.58D-05	5.35D-14	7.58D-05	5.49D-08
4.000D+03	2.95D+01	1.51D-08	2.95D+01	6.51D-05	5.32D-14	6.51D-05	4.72D-08
5.000D+03	2.94D+01	1.69D-08	2.94D+01	5.78D-05	5.29D-14	5.78D-05	4.19D-08
6.000D+03	2.93D+01	1.84D-08	2.93D+01	5.24D-05	5.27D-14	5.24D-05	3.80D-08
7.000D+03	2.91D+01	1.98D-08	2.91D+01	4.82D-05	5.25D-14	4.82D-05	3.49D-08
8.000D+03	2.90D+01	2.11D-08	2.90D+01	4.48D-05	5.23D-14	4.48D-05	3.25D-08
9.000D+03	2.89D+01	2.24D-08	2.89D+01	4.20D-05	5.21D-14	4.20D-05	3.05D-08
1.000D+04	2.88D+01	2.35D-08	2.88D+01	3.97D-05	5.19D-14	3.97D-05	2.87D-08
2.000D+04	2.81D+01	3.25D-08	2.81D+01	2.69D-05	5.05D-14	2.69D-05	1.95D-08
3.000D+04	2.75D+01	3.91D-08	2.75D+01	2.12D-05	4.95D-14	2.12D-05	1.54D-08
4.000D+04	2.70D+01	4.45D-08	2.70D+01	1.79D-05	4.86D-14	1.79D-05	1.30D-08
5.000D+04	2.65D+01	4.91D-08	2.65D+01	1.56D-05	4.78D-14	1.56D-05	1.13D-08
6.000D+04	2.61D+01	5.32D-08	2.61D+01	1.39D-05	4.71D-14	1.39D-05	1.01D-08
7.000D+04	2.58D+01	5.68D-08	2.58D+01	1.26D-05	4.64D-14	1.26D-05	9.17D-09
8.000D+04	2.55D+01	6.01D-08	2.55D+01	1.16D-05	4.59D-14	1.16D-05	8.41D-09
9.000D+04	2.52D+01	6.31D-08	2.52D+01	1.07D-05	4.53D-14	1.07D-05	7.79D-09
1.000D+05	2.49D+01	6.58D-08	2.49D+01	1.00D-05	4.48D-14	1.00D-05	7.26D-09
2.000D+05	2.26D+01	8.59D-08	2.26D+01	6.16D-06	4.07D-14	6.16D-06	4.46D-09
3.000D+05	2.09D+01	9.85D-08	2.09D+01	4.50D-06	3.77D-14	4.50D-06	3.26D-09
4.000D+05	1.96D+01	1.07D-07	1.96D+01	3.54D-06	3.53D-14	3.54D-06	2.57D-09
5.000D+05	1.84D+01	1.14D-07	1.84D+01	2.91D-06	3.32D-14	2.91D-06	2.11D-09
6.000D+05	1.74D+01	1.19D-07	1.74D+01	2.46D-06	3.14D-14	2.46D-06	1.78D-09
7.000D+05	1.65D+01	1.22D-07	1.65D+01	2.11D-06	2.98D-14	2.11D-06	1.53D-09
8.000D+05	1.57D+01	1.25D-07	1.57D+01	1.85D-06	2.83D-14	1.85D-06	1.34D-09
9.000D+05	1.50D+01	1.27D-07	1.50D+01	1.63D-06	2.70D-14	1.63D-06	1.18D-09
1.000D+06	1.43D+01	1.28D-07	1.43D+01	1.45D-06	2.57D-14	1.45D-06	1.05D-09
2.000D+06	9.52D+00	1.25D-07	9.52D+00	6.03D-07	1.71D-14	6.03D-07	4.37D-10
3.000D+06	6.69D+00	1.10D-07	6.69D+00	3.16D-07	1.21D-14	3.16D-07	2.29D-10
4.000D+06	4.84D+00	9.36D-08	4.84D+00	1.84D-07	8.71D-15	1.84D-07	1.33D-10
5.000D+06	3.55D+00	7.79D-08	3.55D+00	1.14D-07	6.40D-15	1.14D-07	8.24D-11
6.000D+06	2.64D+00	6.41D-08	2.64D+00	7.31D-08	4.75D-15	7.31D-08	5.30D-11
7.000D+06	1.98D+00	5.24D-08	1.98D+00	4.83D-08	3.56D-15	4.83D-08	3.50D-11
8.000D+06	1.49D+00	4.26D-08	1.49D+00	3.27D-08	2.69D-15	3.27D-08	2.37D-11
9.000D+06	1.13D+00	3.45D-08	1.13D+00	2.24D-08	2.03D-15	2.24D-08	1.63D-11
1.000D+07	8.60D-01	2.78D-08	8.60D-01	1.56D-08	1.55D-15	1.56D-08	1.13D-11

FORTRAN STOP

LAWRENCE BERKELEY LABORATORY
TECHNICAL INFORMATION DEPARTMENT
UNIVERSITY OF CALIFORNIA
BERKELEY, CALIFORNIA 94720